

THE HELPING HAND digitizer and graphics tablet is essentially two paddles linked together by an arm. As you move the pointer across the board, both paddle potentiometers rotate and their respective digitized values are stored in the same two memory locations normally used by paddles. The states (on/off) of the two buttons mounted on the drawing board are stored in the same memory locations normally used by the paddles' fire buttons. You can assign virtually any function to each button. Since the Helping Hand uses only the memory locations assigned to the paddle potentiometers and their fire buttons, it can be used with any computer that accepts the standard Atari-type paddles.

Operation

The pointer slides easily to any part of the drawing board. If you follow these simple rules you should never have to force it:

- (1) Make sure the "elbow" of the pointer arm is always to the left of the pointer. The only exception is when the pointer is located in the extreme upper left of the board--the elbow is then above and slightly to the right of the pointer.
- (2) It is best to keep the pointer within the rectangular area marked on the board. (If you go beyond this area, be careful, as each pot is designed to "slip" if pushed past its stop. You will then have to recalibrate it.)
- (3) The entire arm should lie flat against the board as you move it about. You can, however, lift it slightly when you need to place a sheet of paper under it.

THE FOLLOWING DIRECTIONS AND PROGRAM LISTINGS ARE FOR THE COMMODORE VIC-20 AND C-64 ONLY. If you are using another computer which accepts the standard Atari-type paddles, you must first determine where in memory the paddle potentiometer values and fire button states are stored. The Commodore VIC-20 and 64 display 22 and 40 characters on a line, respectively. You will have to tailor each program to fit your computer's screen display. For the drawing programs you will also need to know the screen and color memory locations. The program listings are thoroughly documented, so you should have little trouble converting them to run on your computer.

Connect The Helping Hand to the 9-pin game port of your computer (use port 1 on the C-64). RUN the following one-line program for the VIC-20:

```
10 PRINT PEEK(36872),PEEK(36873): GOTO10
```

or for the C-64:

```
10 PRINT PEEK(54297),PEEK(54298): GOTO10
```

This routine reads the two memory locations in which the values of the paddle potentiometers are stored. The left column of numbers scrolling up the screen represent the digitized values generated by the potentiometer mounted in the base of the arm (the X pot). Similarly, the right column is generated by the potentiometer mounted in the elbow (the Y pot). These numbers should change as you move the pointer around the board. Neither potentiometer will move over the full range of values (0-255) as long as the pointer is confined to move within the rectangular area.

Calibration

Before you can use any of the demonstration programs, the Helping Hand must be calibrated. RUN the one-line routine again and move the pointer so that it's pointing to the top left corner of the rectangular boarder (the elbow will be off the board, above and to the right of the pointer). Watch the left column of numbers on the screen. They should gradually decrease as you move towards that corner and not quite reach zero when you get to that corner--anywhere between 3 and 10 is acceptable. If the numbers increase instead, the leads from the X pot are reversed. Trace the wire from the X pot to the terminal block on the underside and reverse the leads there. The X pot itself needs to be adjusted if the left column of numbers reached 3-10 well before the the pointer got to the top left corner or if they had not yet reached 3-10 by then. To make this adjustment, loosen the screw that holds the shaft of the X pot to the arm (a fraction of a turn will do). With the pointer at the upper left corner of the rectangular border, use a pair of long-nose pliers to turn the shaft of the X pot until the left column of numbers are between 3 and 10. Hold the shaft in this position with the pliers and tighten the screw. The Y pot is adjusted in a similar manner. Move the pointer arm away from the top left corner along the line that runs across the top of the board, until the pointer is touching the base of the arm, i.e. the arm is fully collapsed. Watch the right column of numbers and they should just reach zero as the pointer gets to the base. If the numbers increase, the Y pot's leads need to be reversed at the terminal block. If they decrease, but reach zero too soon or don't reach zero when the arm is fully collapsed, adjust the Y pot using the long-nose pliers.

Type in and SAVE a copy of the Calibration program. RUN the program and follow the instructions on the screen . Use the four computed values: A1, B1, A2, B2 in place of those found in the demonstration programs. For the VIC-20 and C-64 the range of values are: A1 and A2:.015-.030 ; B1:2-3 ; B2:.4-.6 . The four values are used in the conversion of the digitized values to screen and drawing board locations. Remember, the best performance is obtained if (a) the X pot reads 3-10 when the pointer is at the upper left corner, and (b) the Y pot reads zero when the arm is fully collapsed.

You can help us by sending in your comments.

Persimmon Peripherals
Route 2, Box 2306A
Clayton, GA 30525

```

100 REM **      CALIBRATION OF DIGITIZER FOR VIC20      **
110 REM **                                             **
120 REM **REFERENCE: BYTE MAGAZINE FEB.'82 P6.72-86 **
130 REM **                                             **
140 PRINTCHR$(147);REM CLEAR THE SCREEN
150 PI=3.14159265
160 R=7;REM LENGTH OF EACH ARM (INCHES)
170 DEF FNAS(X)=ATN(X/SQR(-X*X+1)); REM ARCSINE FUNCTION
180 PRINT"SET THE POINTER TO THE CALIBRATION POINT LOCATED UPPER LEFT"
190 PRINT:PRINT"HIT [RETURN] WHEN SET"
200 GETA$:IFA$=""THEN200
210 X1=-3:Y1=-1;REM BOARD COORDINATES OF CALIBRATION POINT
220 QX=PEEK(36872):QY=PEEK(36873);REM READ PADDLES
230 F1=ATN(Y1/X1)+PI
240 H1=SQR(X1*X1+Y1*Y1)
250 S2=2*FNAS(H1/(2*R))
260 S1=F1-PI/2+S2/2
270 PRINTCHR$(147);PRINT"SET THE POINTER TO THE CALIBRATION POINT LOCATED LOWER RIGHT"
280 PRINT:PRINT"HIT [RETURN] WHEN SET"
290 GETA$:IFA$=""THEN290
300 X2=7:Y2=-10;REM BOARD COORDINATES OF CALIBRATION POINT
310 RX=PEEK(36872):RY=PEEK(36873);REM READ PADDLES
320 F2=2*PI+ATN(Y2/X2)
330 H2=SQR(X2*X2+Y2*Y2)
340 T2=2*FNAS(H2/(2*R))
350 T1=F2-PI/2+T2/2
360 A1=(S1-T1)/(QX-RX);REM TAPER FOR PADDLE X
370 A2=(S2-T2)/(QY-RY);REM INCREMENT FOR PADDLE X
380 B1=(QX*T1-RX*S1)/(QX-RX);REM TAPER FOR PADDLE Y
390 B2=(QY*T2-RY*S2)/(QY-RY);REM INCREMENT FOR PADDLE Y
400 PRINT:PRINT"A1= "A1
410 PRINT"B1= "B1
420 PRINT"A2= "A2
430 PRINT"B2= "B2
440 PRINT:PRINT"CALIBRATION COMPLETED"
450 REM
460 REM USE THE ABOVE FOUR CALIBRATION VALUES IN THE DRAWING PROGRAMS
470 REM
READY.

```

```

100 REM **CALIBRATION OF DIGITIZER FOR 64 **
110 REM **
120 REM **REF: BYTE FEB.'82 PG.72-86 **
130 REM **
140 REM **
150 PRINTCHR$(147):REM CLEAR THE SCREEN
160 GOSUB1030:REM POKE PADDLE-READING ROUTINE INTO MEMORY
170 PI=3.14159265
180 R=7:REM LENGTH OF EACH ARM (INCHES)
190 DEF FNAS(X)=ATN(X/SQR(-X*X+1)): REM ARCSINE FUNCTION
200 PRINT"SET THE POINTER TO THE CALIBRATION POINT LOCATED UPPER LEFT"
210 PRINT:PRINT"HIT [RETURN] WHEN SET"
220 X1=-3:Y1=-1:REM BOARD COORDINATES OF CALIBRATION POINT
230 GETA$:IFA$=""THEN320
240 SYSPX:RX=PEEK(251):SYSPY:RY=PEEK(251): REM READ PADDLES
250 F1=ATN(Y1/X1)+PI
260 H1=SQR(X1*X1+Y1*Y1)
270 S2=2*FNAS(H1/(2*R))
280 S1=F1-PI/2+S2/2
290 PRINTCHR$(147)
300 PRINT"SET THE POINTER TO THE CALIBRATION POINT LOCATED LOWER RIGHT"
310 PRINT:PRINT"HIT [RETURN] WHEN SET"
320 GETA$:IFA$=""THEN320
330 X2=7:Y2=-10:REM BOARD COORDINATES OF CALIBRATION POINT
340 SYSPX:RX=PEEK(251):SYSPY:RY=PEEK(251):REM READ PADDLES
350 F2=2*PI+ATN(Y2/X2)
360 H2=SQR(X2*X2+Y2*Y2)
370 T2=2*FNAS(H2/(2*R))
380 T1=F2-PI/2+T2/2
390 A1=(S1-T1)/(QX-RX):REM TAPER FOR PADDLE X
400 A2=(S2-T2)/(QY-RY):REM INCREMENT FOR PADDLE X
410 B1=(QX*T1-RX*S1)/(QX-RX):REM TAPER FOR PADDLE Y
420 B2=(QY*T2-RY*S2)/(QY-RY):REM INCREMENT FOR PADDLE Y
430 PRINT:PRINT"A1= "A1
440 PRINT"B1= "B1
450 PRINT"A2= "A2
460 PRINT"B2= "B2
470 PRINT:PRINT"CALIBRATION COMPLETED"
480 REM
490 REM USE THESE FOUR CALIBRATION VALUES IN LINE 280 OF THE DRAW PROGRAM
500 END
1000 REM**
1010 REM**PADDLE-READING ROUTINE**
1020 REM**
1030 PX=49152:FORI=0TO16:READA:POKEPX+I,A:NEXT
1040 DATA 169,0,170,168,24,109,25,212,144,1,200,202,208,247,132,251,96
1050 PY=49152:FORI=0TO16:READA:POKEPY+I,A:NEXT
1060 DATA 169,0,170,168,24,109,26,212,144,1,200,202,208,247,132,251,96
1070 RETURN
READY.

```

```

100 REM **LD-RES DRAW WITH SOUND FOR VIC20**
110 REM **WORKS WITH ANY SIZE MEMORY **
120 REM **
130 REM **
140 REM **HOLD DOWN THE RIGHT SWITCH TO TURN SOUND ON OR OFF **
150 REM **HOLD DOWN THE LEFT SWITCH TO CLEAR THE SCREEN **
160 REM **
170 REM **TO CHANGE THE DISPLAYED CHARACTER PRESS ANY KEY ON**
180 REM **THE KEYBOARD... EXCEPT FOR THE NUMBERS 1 TO 9 **
190 REM **
200 REM **TO CHANGE THE COLOR OF THE DISPLAYED CHARACTER **
210 REM **PRESS A NUMBERED KEY (1-9)...DON'T USE 'CTRL' !! **
220 REM **
230 REM **
240 REM **
250 IF PEEK(44)=18 THEN EV=1: REM TEST FOR EXPANDED VIC
260 SM=4096-3584*(EV=0): REM START OF SCREEN MEMORY
270 PRINT CHR$(147): REM CLEAR THE SCREEN
280 POKE 37139,0: DD=37154: PA=37137: PB=37152: VL=36878: R=7: PI=3.14159265
290 SC=30: CC=3: POKE VL,0: REM DISPLAY CYAN ARROWS; TURN SOUND OFF
300 A1=.0249: B1=2.071: A2=.0241: B2=.359: REM INSERT YOUR CALIBRATION VALUES HERE
310 RX=PEEK(36872): RY=PEEK(36873): REM READ PADDLES
320 MU=100+RX+RY: REM VALUE OF MUSICAL NOTE
330 IF MU>245 THEN MU=245
340 IF MU<128 THEN MU=128
350 POKE 36875,MU: REM PLAY NOTE (IF SOUND ON)
360 GET A$: IF A$="" THEN 390: REM SKIP NEXT TWO LINES IF NO KEY IS PRESSED
370 IF VAL(A$)>0 AND VAL(A$)<10 THEN CC=VAL(A$)-1: GOTO 390: REM CHANGE CHARACTER COLOR
380 GOSUB 3000: REM COMPUTE SCREEN CODE OF NEW CHARACTER
390 GOSUB 1030: REM CONVERT PADDLE VALUES TO BOARD COORDINATES (-7<X<7;-10<Y<0)
400 GOSUB 2000: REM READ SWITCHES
420 IF SY=1 THEN PRINT CHR$(147): REM CLEAR THE SCREEN
430 ROW=INT(ABS(Y*22/10))
440 COLUMN=INT(ABS((X+7)*21/14)): IF COLUMN>21 THEN COLUMN=21
450 IF SX=0 THEN POKE SL,SC: POKE CL,CC: FORT=1 TO 50: NEXT T: POKE SL,32: FORT=1 TO 10: NEXT T
460 SL=SM+22*RO+CO: CL=SL+30720-3072*(EV=1): REM SCREEN AND COLOR LOCATIONS
470 IF SX=1 THEN POKE SL,SC: POKE CL,CC: SO=SL: REM DISPLAY CHARACTER
480 GOTO 310
1000 REM **
1010 REM **SUBROUTINES**
1020 REM **
1030 TI=A1*RX+B1: TY=A2*RY+B2
1040 H=2*R*SIN(TY/2)
1050 THETA=TX+PI/2-TY/2
1060 X=H*COS(THETA): Y=H*SIN(THETA)
1070 RETURN
2000 POKEDD,127: SY=-((PEEK(PB)AND 128)=0): POKEDD,255
2010 SX=-((PEEK(PA)AND 16)=0): RETURN
3000 AS=ASC(A$): IF AS=222 THEN SC=94: RETURN
3010 AS=AS+64*(AS>224)+96*(AS>191) AND AS<224)
3020 SC=AS-32-32*(AS=>32) AND AS<63)+32*(AS=>161) OR AS=>64 AND AS<95)
3030 RETURN
READY.

```

```

100 REM **LO-RES DRAW**
110 REM ** FOR C64 **
120 REM **
130 REM **
140 REM **HOLD DOWN LEFT SWITCH TO CLEAR THE SCREEN**
150 REM **
160 REM **TO CHANGE THE DISPLAYED CHARACTER PRESS ANY KEY**
170 REM **ON THE KEYBOARD... EXCEPT FOR THE NUMBERS (1-9)**
180 REM **
190 REM **TO CHANGE THE COLOR OF THE DISPLAYED CHARACTER **
200 REM **PRESS A NUMBERED KEY (1-9)...DON'T USE 'CTRL'!**
210 REM **
220 REM **
230 REM **
240 GOSUB 4000:REM POKE PADDLE-READING ROUTINE INTO MEMORY
250 PRINTCHR$(147):REM CLEAR THE SCREEN
260 R=7:PI=3.14159265
270 SC=30:CC=3:REM DISPLAY CYAN ARROWS
280 A1=.0249:B1=2.071:A2=.0241:B2=.359:REM INSERT CALIBRATION VALUES HERE
290 SYSPX:RX=PEEK(251):SYSPY:RY=PEEK(251):REM READ PADDLES
300 GETA$:IFA$=""THENJ30:REM SKIP NEXT TWO LINES IF NO KEY PRESSED
310 IFVAL(A$)>0ANDVAL(A$)<10THENCC=VAL(A$)-1:GOTOJ30:REM CHANGE CHR COLOR
320 GOSUB3000:REM COMPUTE SCREEN CODE OF NEW CHARACTER
330 GOSUB1030:REM CONVERT PADLE VALUES TO BOARD COORDINATES (-7<X<7;-10<Y<0)
340 GOSUB 2000:REM READ RIGHT-HAND SWITCH
350 IFSY=1THEN PRINTCHR$(147):REM CLEAR THE SCREEN
360 ROW=INT(ABS(Y*24/10)):IFROW>24THENROW=24
370 COLUMN=INT(ABS((X+7)*39/14)):IFCOLUMN>39THENCOLUMN=39
380 REM TO MOVE SINGLE CHARACTER AROUND SCREEN ADD THIS LINE HERE: POKESL,32
390 SL=1024+C0+40*R0:POKESL,SC:POKESL+54272,CC:S0=SL:REM DISPLAY CHAR ON SCR
400 GOTO290

1000 REM **
1010 REM **SUBROUTINES**
1020 REM **
1030 TX=A1*RX+B1:TY=A2*RY+B2
1040 H=2*R*SIN(TY/2)
1050 THETA=TX+PI/2-TY/2
1060 X=H*COS(THETA):Y=H*SIN(THETA)
1070 RETURN

2000 SY=-((PEEK(56321)AND8)=0):RETURN
3000 AS=ASC(A$):IFAS=222THENSC=94:RETURN
3010 AS=AS+64*(AS>224)+96*(AS>191ANDAS<224)
3020 SC=AS-32-32*(AS>32ANDAS<63)+32*(AS>161ORAS>64ANDAS<95)
3030 RETURN

4000 PX=49152:FORI=0TO16:READA:POKEPX+I,A:NEXT
4010 DATA 169,0,170,168,24,109,25,212,144,1,200,202,208,247,132,251,96
5000 PY=49169:FORI=0TO16:READA:POKEPY+I,A:NEXT
5010 DATA 169,0,170,168,24,109,26,212,144,1,200,202,208,247,132,251,96
5020 RETURN

READY.

```

```

100 REM**HI-RES DRAW**
110 REM**FOR VIC20 **
120 REM**      **
130 REM**  REQUIRES **
140 REM**    SUPER **
150 REM**  EXPANDER **
160 REM**      **
170 REM**      **
180 REM**HOLD DOWN THE RIGHT SWITCH TO CHANGE THE COLOR **
190 REM**HOLD DOWN THE LEFT SWITCH TO CLEAR THE SCREEN **
200 REM**      **
210 REM**      **
220 REM**      **
230 POKE37139,0:DD=37154:PA=37137:PB=37152:CC=6:X1=0:Y1=0:R=7:PI=3.14159265
240 A1=.0249:B1=2.071:A2=.0241:B2=.359:REM INSERT YOUR CALIBRATION VALUES HERE
250 GRAPHIC2:COLOR11,6,0,6
260 RX=PEEK(36872):RY=PEEK(36873):REM READ PADDLES
270 GOSUB2000:REM CONVERT PADDLE VALUES TO BOARD COORDINATES (-7<X2<7;-10<Y2<0)
280 X2=(X2+7)*(1023/14):Y2=- (Y2*(1023/10)):REM CONVERT TO GRAPHIC SCREEN COORDINATES
290 IFX2<0THENX2=0
300 IFY2<0THENY2=0
310 GOSUB1030:REM READ SWITCHES
320 IFSY=1THEN:SCNCLR:GOTO260
330 IFSX=1THENCC=CC+1+9*(CC=8):REGIONCC:REM CHANGE DRAWING COLOR
340 DRAW2,X1,Y1TOX2,Y2:X1=X2:Y1=Y2:REM DRAW A LINE FROM PREVIOUS POINT TO NEW POINT
350 REM TO DRAW WITH POINTS INSTEAD OF LINES, REPLACE ABOVE LINE WITH: POINT 2,X2,Y2
360 GOTO260
1000 REM**      **
1010 REM**SUBROUTINES**
1020 REM**      **
1030 POKEDD,127:SY=-((PEEK(PB)AND128)=0):POKEDD,255
1040 SX=-((PEEK(PA)AND16)=0):RETURN
2000 TX=A1*RX+B1:TY=A2*RY+B2
2020 H=2*R*SIN(TY/2)
2050 THETA=TX+PI/2-TY/2
2060 X2=H*COS(THETA):Y2=H*SIN(THETA):RETURN
READY.

```

```

100 REM **HI-RES DRAW**
110 REM ** FOR C-64 **
120 REM **
130 REM **
140 REM **REQUIRES ABACUS GRAPHICS 64 SOFTWARE **
150 REM **
160 REM **
170 REM **HOLD DOWN LEFT SWITCH TO CLEAR THE SCREEN **
180 REM **
190 REM **TO CHANGE THE COLOR OF THE LINE BEING DRAWN**
200 REM **PRESS A NUMBERED KEY (1-8)...DON'T USE CTRL**
210 REM **
220 REM **
230 REM **
240 GOSUB4000: REM POKE PADDLE-READING ROUTINE INTO MEMORY
250 HIRES7,7:CC=2: REM SET UP BLUE HI-RES SCREEN AND LINES WILL BE WHITE
260 R=7:PI=3.14159265
270 REM: NEXT LINE CONVERTS FROM KEYBOARD VALUES TO PEEK(197) VALUES
271 C(1)=56:C(2)=59:C(3)=8:C(4)=11:C(5)=16:C(6)=19:C(7)=24:C(8)=27:C(9)=32
280 A1=.0249:B1=2.071:A2=.0241:B2=.359:REM INSERT CALIBRATION VALUES HERE
290 SYSPX:RX=PEEK(251):SYSPY:RY=PEEK(251):REM READ PADDLES
330 GOSUB1030:X1=(X+5)*31.9:Y1=(Y+10)*19.9:REM PADDLE VALUES TO SCREEN LOCATIONS
340 GOSUB2000: REM READ LEFT-HAND SWITCH
350 IFSY=1THEN HIRES7,7: REM CLEAR SCREEN
355 IFX1<0THENX1=0: REM THESE FOUR LINES KEEP DRAWING WITHIN BOUNDS
356 IFY1<0THENY1=0
357 IFX1>319THENX1=319
358 IFY1>199THENY1=199
359 CA=PEEK(197):IFCA<>64THENGOSUB3000:REM COMPUTE NEW DRAWING COLOR
360 DRAWX1,Y1,X2,Y2,CC:X2=X1:Y2=Y1
400 GOTO 290
1000 REM**
1010 REM**SUBROUTINES**
1020 REM**
1030 TX=A1*RX+B1:TY=A2*RY+B2
1040 H=21R*SIN(TY/2)
1050 THETA=TX+PI/2-TY/2
1060 X=H*COS(THETA):Y=H*SIN(THETA)
1070 RETURN
2000 SY=-((PEEK(56321)AND8)=0):RETURN
3000 FOR TC=1TO9:IFCA<>C(TC) THEN NEXT
3010 CC=TC: RETURN
4000 PX=49152:FORI=0TO16:READA:POKEPX+I,A:NEXT
4010 DATA 169,0,170,168,24,109,25,212,144,1,200,202,208,247,132,251,96
5000 PY=49169:FORI=0TO16:READA:POKEPY+I,A:NEXT
5010 DATA 169,0,170,168,24,109,26,212,144,1,200,202,208,247,132,251,96
5020 RETURN

```

READY.